Spin 1/2 in an external magnetic field

Matrix representation of the spin components

In[1]:=
```
MatrixForm[σx = {{0, 1}, {1, 0}}]
```
Out[1]//MatrixForm=

$$\begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}$$

In[2]:=
```
MatrixForm[σy = {{0, -I}, {I, 0}}]
```
Out[2]//MatrixForm=

$$\begin{pmatrix} 0 & -i \\ i & 0 \end{pmatrix}$$

In[3]:=
```
MatrixForm[σz = {{1, 0}, {0, -1}}]
```
Out[3]//MatrixForm=

$$\begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}$$

In[4]:=
```
Sx = (ℏ / 2) * σx; Sy = (ℏ / 2) * σy; Sz = (ℏ / 2) * σz;
```

Matrix representation of the energy.

In classical physics the energy of a magnetic moment $\vec{\mu}$is given by E=-$\vec{\mu}.\vec{B}$, where $\vec{B}$is the external magnetic field in which the magnetic moment is placed and $\vec{\mu}$=$\gamma\vec{s}$, with γ being the gyromagnetic constant. The components of a quantum spin are represented by matrices and its energy is represented by a matrix whose eigenvalues are the possible energies the spin can have.

Magnetic field :

In[5]:=
```
MatrixForm[Bvec = B {nx, ny, nz}]
```
Out[5]//MatrixForm=

$$\begin{pmatrix} B\,nx \\ B\,ny \\ B\,nz \end{pmatrix}$$

Magnetic moment :

In[6]:=
```
MatrixForm[μx = γ Sx]
```
Out[6]//MatrixForm=

$$\begin{pmatrix} 0 & \frac{\gamma \hbar}{2} \\ \frac{\gamma \hbar}{2} & 0 \end{pmatrix}$$

In[7]:=
```
MatrixForm[μy = γ Sy]
```
Out[7]//MatrixForm=

$$\begin{pmatrix} 0 & -\frac{1}{2} i \gamma \hbar \\ \frac{i \gamma \hbar}{2} & 0 \end{pmatrix}$$

In[8]:=
```
MatrixForm[μz = γ Sz]
```
Out[8]//MatrixForm=

$$\begin{pmatrix} \frac{\gamma \hbar}{2} & 0 \\ 0 & -\frac{\gamma \hbar}{2} \end{pmatrix}$$

The energy is represented by a matrix (“Hamiltonian”):

In[9]:=
```
MatrixForm[H = FullSimplify[-B (μx * nx + μy * ny + μz * nz)]]
```
Out[9]//MatrixForm=

$$\begin{pmatrix} -\frac{1}{2} B\, nz\, \gamma \hbar & -\frac{1}{2} B\, (nx - i\, ny)\, \gamma \hbar \\ -\frac{1}{2} B\, (nx + i\, ny)\, \gamma \hbar & \frac{1}{2} B\, nz\, \gamma \hbar \end{pmatrix}$$

The possible values of the energy do not depend on the direction of the magnetic field and correspond, respectively, to the orientations parallel and antiparallel of $\vec{s}$ and $\vec{B}$:

In[10]:=
```
Simplify[Eigenvalues[H], Assumptions → {nx^2 + ny^2 + nz^2 == 1}]
```
Out[10]= $\left\{-\frac{1}{2} B \gamma \hbar, \frac{B \gamma \hbar}{2}\right\}$

Polarization states, spinors

The two normalized eigenvectors of **H**, with $\vec{B}$ in the direction of the x-, y-, z-axis are, respectively, matrix representations (“spinors”) of the polarization states parallel and antiparallel to the x-, y-, z-axis. Here the eigenvector corresponding to the eigenvalue E=-γBℏ/2 represents the “parallel” polarization and the one corresponding to the eigenvalue E=γBℏ/2 the “antiparallel” polarization. Due to the construction of the **H**-matrix, these polarization states are represented by th eigenvectors of the Pauli matrices corresponding, respectively, to the eigenvalues -1,1:

In[11]:=
```
Eigenvalues[σx]
```
Out[11]= {-1, 1}

In[12]:=
```
Map[Normalize, Eigenvectors[σx]]
```
Out[12]= $\left\{\left\{-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}\right\}$

In[13]:=
```
Eigenvalues[σy]
```
Out[13]= {-1, 1}

In[14]:=
```
Map[Normalize, Eigenvectors[σy]]
```
Out[14]= $\left\{\left\{\frac{i}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \left\{-\frac{i}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}\right\}$

In[15]:=
```
Eigenvalues[σz]
```
Out[15]=
```
{-1, 1}
```

In[16]:=
```
Map[Normalize, Eigenvectors[σz]]
```
Out[16]=
```
{{0, 1}, {1, 0}}
```

In[17]:=

Random polarization and mean value

Chose a magnetic field pointing along the z-axis, such that

In[18]:=
```
MatrixForm[Hz = H /. {nx → 0, ny → 0, nz → 1}]
```
Out[18]//MatrixForm=

$$\begin{pmatrix} -\frac{1}{2} B \gamma \hbar & 0 \\ 0 & \frac{B \gamma \hbar}{2} \end{pmatrix}$$

The eigenvalues are (independent of the direction of the magnetic field)

In[19]:=
```
EigenvaluesHz = Eigenvalues[Hz]
```
Out[19]=

$$\left\{-\frac{1}{2} B \gamma \hbar, \frac{B \gamma \hbar}{2}\right\}$$

and the normalized eigenvectors $\{E_1, E_2\}$ are

In[20]:=
```
χbasis = Map[Normalize, Eigenvectors[Hz]]
```
Out[20]=
```
{{1, 0}, {0, 1}}
```

Construct *an arbitrary spinor* as superposition of the above basis spinors, where $|\alpha|^2 + |\beta|^2 = 1$

In[21]:=
```
χ = α * χbasis[[1]] + β * χbasis[[2]]
```
Out[21]=
```
{α, β}
```

and compute the quadratic form $\boldsymbol{\chi}^{\dagger}.\mathbf{H}.\boldsymbol{\chi}$

In[22]:=
```
Simplify[Conjugate[χ].Hz.χ] // TraditionalForm
```
Out[22]//TraditionalForm=

$$\frac{1}{2} B \gamma \hbar (\beta \beta^* - \alpha \alpha^*)$$

The quadratic form can be written as $\boldsymbol{\chi}^{\dagger}.\mathbf{H}.\boldsymbol{\chi} = \alpha \alpha^* E_1 + \beta \beta^* E_2$,

In[23]:=
```
Conjugate[α] α * EigenvaluesHz[[1]] + Conjugate[β] β * EigenvaluesHz[[2]] //
  Simplify // TraditionalForm
```
Out[23]//TraditionalForm=

$$\frac{1}{2} B \gamma \hbar (\beta \beta^* - \alpha \alpha^*)$$

This shows that $\alpha \alpha^* = |\alpha|^2$ and $\beta \beta^* = |\beta|^2$ can be interpreted, respectively, as probabilities for the parallel and antiparallel polarization with respect to the magnetic field pointing along the z-axis, and the quadratic form $\boldsymbol{\chi}^{\dagger}.\mathbf{H}.\boldsymbol{\chi}$ as the mean value of the polarization. For the choices {1,0} and {0,1}, respectively, for $\{\alpha,\beta\}$ the corresponding spinor $\boldsymbol{\chi}$ represents the "pure states" for polarization parallel and antiparallel to the z axis.

The same reasoning can be used by choosing a Hamiltonian where the magnetic field points into another, arbitrary direction. The reason is that the corresponding spinors form always an orthogonal basis (show this as an exercise). Chose for example the x-axis :

In[24]:=
```
MatrixForm[Hx = H /. {nx → 1, ny → 0, nz → 0}]
```

Out[24]//MatrixForm=

$$\begin{pmatrix} 0 & -\frac{1}{2} B \gamma \hbar \\ -\frac{1}{2} B \gamma \hbar & 0 \end{pmatrix}$$

The eigenvalues are again

In[25]:=
```
EigenvaluesHx = Eigenvalues[Hx]
```

Out[25]= $\left\{-\frac{1}{2} B \gamma \hbar, \frac{B \gamma \hbar}{2}\right\}$

and the normalized eigenvectors $\{E_1, E_2\}$ are here

In[26]:=
```
χbasis = Map[Normalize, Eigenvectors[Hx]]
```

Out[26]= $\left\{\left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \left\{-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}\right\}$

Construct again *an arbitrary spinor* as superposition of the above basis spinors, where $|\alpha|^2 + |\beta|^2 = 1$

In[27]:=
```
χ = α * χbasis[[1]] + β * χbasis[[2]]
```

Out[27]= $\left\{\frac{\alpha}{\sqrt{2}} - \frac{\beta}{\sqrt{2}}, \frac{\alpha}{\sqrt{2}} + \frac{\beta}{\sqrt{2}}\right\}$

The quadratic form $\boldsymbol{\chi}^{\dagger}.\mathbf{H}.\boldsymbol{\chi}$ is the same as for the case where the field points along the z-axis,

In[28]:=
```
Simplify[Conjugate[χ].Hx.χ] // TraditionalForm
```

Out[28]//TraditionalForm=

$$-\frac{1}{2} B \gamma \hbar (\alpha \alpha^* - \beta \beta^*)$$

and the probabilistic interpretation as a mean value is the same.